

# 한 윈도우에 다중 플롯을 하려면...

## IDL이 지원하는 그래픽스의 종류

IDL은 현재 총 4 종류의 그래픽 시스템을 지원하고 있습니다. IDL 1.0과 함께 지속되어온 Direct Graphics (이하 D·G), IDL 5.0부터 추가된 Object Graphics(이하 O·G), 그리고 각각 IDL 6.0과 8.0부터 추가된 iTools와 New Graphics(이하 N·G)가 그것입니다.

각각은 모두 다른 특징을 가지고 있습니다. IDL의 시작과 함께한 D·G는 사용법이 간단하고, 신속하게 결과물을 표현할 수 있어서 현재까지 가장 많은 사용자 층을 가지고 있습니다.

O·G는 그래픽 품질이 우수합니다. 하지만 제대로 다루기는 조금 어렵다는 단점이 있습니다. 정확히 이해하고 다루기 위해서 많은 노력과 시간을 필요로 합니다. 한 장 강의에서도 다루기에는 양이 너무 많아서 제외하였습니다.

iTools와 N·G는 D·G와 O·G의 중간 단계라 할 수 있습니다. 사용하기에 쉽지만 고품질의 그래픽을 표출할 수 있습니다. 결과물만 보면 큰 차이가 없습니다. 다만 iTools는 프로시저, N·G는 함수이며 문법에 다소 차이가 있을 뿐입니다.


## 각 그래픽 시스템의 기본적인 표출방법

우선은 각 그래픽스의 기본적인 명령어와 그에 대한 결과물을 비교해보도록 하겠습니다. D·G, N·G, iTools에서 sine함수를 표현하기 위한 문법 및 결과입니다.

```
x = indgen(361)
y = sin(x * !dtr)
xtitle = 'X-Title' & ytitle = 'Y-Title'
```

```
plot, x, y, xtitle=xtitle, ytitle=ytitle, charsize = 2.5
p = plot(x, y, xtitle=xtitle, ytitle=ytitle, $
font_size = 25)
```

```
iplot, x, y, xtitle=xtitle, ytitle=ytitle, font_size = 25
```



우선 그래픽 상에서 눈에 띄는 것은 바탕색입니다. 물론 background라는 키워드 혹은 !p.background라는 시스템 변수 변경을 통해 D·G에서의 바탕도 하얗게 표현할 수 있지만, 기본적으로 D·G는 검은 바탕을 사용합니다. 또한 x, y 축의 제목과 숫자들이 N·G와 iplot에 비해서 딱딱하고 직선적이라는 느낌이 있습니다.

명령어 부분에서도 차이점은 보입니다. D·G에서는 문자의 크기 변경을 위해 charsize란 키워드를 사용했지만, N·G와 iplot에서는 font\_size란 키워드를 사용합니다(기본적으로 IDL의 키워드는 상당히 직관적이기 때문에 유사한 키워드가 많이 있습니다만, 간혹 차이가 있으니 help를 통해 확인을 해보시기 바랍니다).

가장 큰 차이점으로 D·G의 경우 윈도우 안에 오직 그래프만이 표현되었지만, N·G와 iTools에는 그래프 외에 프린트, 디스켓, 연필 등의 아이콘을 볼 수 있습니다. 그렇습니다. N·G와 iTools에서는 결과물을 프린트, 저장, 수정 등을 할 수 있습니다. 아주 쉽게 말이죠.

## 다중 플롯을 해보자!

사실 단일 플롯은 IDL 유저라면 쉽게 알 수 있는 부분입니다. 하지만 우리는 보고서나 논문에서 그래프 하나만을 표현하는 경우가 거의 없습니다. Before & After 혹은 시계열적인 자료를 표출하는 경우가 대부분입니다. 언급한 3가지 그래픽 시스템에서 다중 플롯을 어떻게 적용할 수 있을까요?

### - D·G에서의 표출방법



D·G에서의 표출 방법은 2가지가 있습니다.

'1. !p.multi를 이용하여 윈도우를 분할하기'

'2. 각 plot 프로시저 사용 시 position 키워드로 각각의 위치 지정하기'가 그것입니다. 예문을 통해 보면 아래와 같습니다.

```
!P.MULTI = [0, 2, 2]
plot, x, y
plot, x
!P.MULTI = [1, 1, 2]
plot, y, x
```

2x2 분할로 표현하고자 했다면 총 4번의 plot을 하면



될 것입니다. 하지만 특정 그래프를 크게 표현하여 강조하고자 한다면 추가적인 명령을 필요로 하는데 바로 !p.multi라는 변수의 중복 사용입니다. 위의 경우 !p.multi 이후 3개의 인수만을 입력하였는데 최대 5개의 인수를 입력할 수 있습니다. 숫자가 의미하는 것은 ①추가할 수 있는 plot의 수, ②,③,④는 각각 행, 열, z축으로의 개수, ⑤표출 방향입니다(3개의 숫자로만 표현하는 것은 ④, ⑤를 생략하는 경우입니다).


⑥의 표출방향은 0 또는 1로 표현할 수 있으며, 디폴트(0)일 경우 그래프의 표출은 왼쪽 상단을 기준으로 좌→우, 상→하의 순서로, 1일 경우에는 상→하, 좌→우의 순서로 이루어집니다.

①에서 0을 지정했다면 행×열에서 왼쪽 상단에 플롯을 하게 되고, 1이라면 앞으로 1개의 플롯을 할 수 있단 의미이므로 가장 마지막 지점인 오른쪽 하단에 표출을 하게 됩니다. 그러므로 !p.multi=[0,2,2]는 '2×2 행렬로 표현하되 왼쪽 상단에', !p.multi=[1,1,2]는 '1×2 행렬의 아래쪽에' 표출을 하란 의미를 갖게 됩니다.

또 다른 방법으로는 position 키워드를 이용하는 것으로, 다음과 같이 사용할 수 있습니다.

```
position1 = [0.05, 0.55, 0.45, 0.95]
position2 = [0.55, 0.55, 0.95, 0.95]
position3 = [0.05, 0.05, 0.95, 0.45]
plot, x, y, position = position1
plot, x, position = position2, /noerase
plot, y, x, position = position3, /noerase
```

첫 번째 방법에 비해 다소 손이 많이 가지만, 세세한 위치 조절이 가능하다는 장점이 있습니다. 두 경우 모두 아래와 같이 표현이 될 것입니다.



### - N·G 및 iTools에서의 표출방법


얼마 전 출시된 IDL 8.2 SP1(8.2.1)의 What's new에서는 N·G에 추가된 기능을 소개하고 있습니다. layout이란 키워드를 사용하여 다중 플롯 시 위치를 쉽게 조절하는 방법입니다(한장 강의 주제를 다중 플롯으로 설정한 이유이기도 합니다). 아래의 예문과 같이 사용합니다.

```
p1 = plot(x, y, LAYOUT=[2,2,1])
p2 = plot(x, LAYOUT=[2,2,2], /current)
```

```
p3 = plot(y, x, LAYOUT = [1,2,2], /current)
```

D·G의 !p.multi와 유사하지만 대괄호 안의 순서는 차이를 보입니다. [행의 수, 열의 수, 위치]를 의미합니다. 또한, D·G에서와 동일하게 position 키워드를 이용하는 방법도 있으며 다음의 예문과 같이 표현하게 됩니다. 결과는 layout을 이용한 위의 경우와 동일합니다.


```
p1 = plot(x, y, position = position1)
p2 = plot(x, position = position2, /current)
p3 = plot(y, x, position = position3, /current)
```



한편, 이 방법은 iTools에서도 사용할 수 있습니다. layout, position을 이용하는 경우의 문법은 아래와 같으며, 함수/프로시저의 차이점만 빼면 N·G와 동일합니다. 확인 후, 자신의 목적에 맞는 그래픽 시스템을 이용하시면 됩니다.

```
iplot, x, y, LAYOUT = [2, 2, 1]
iplot, x, LAYOUT = [2, 2, 2], /current
iplot, y, x, LAYOUT = [1, 2, 2], /current
```

```
iplot, x, y, position = position1
iplot, x, position = position2, /current
iplot, y, x, position = position3, /current
```



## 8.2 SP1 의 새 기능

layout을 이용한 다중 플롯 외에 8.2 SP1에서 추가된 기능은 더 있습니다. 그래픽 관련된 내용이 많지만, 다른 부분에 대한 내용도 포함하고 있습니다. 이에 대한 자세한 내용은 다음의 페이지에서 확인하실 수 있습니다.

[http://idl-envi.co.kr/callcenter/tips/file/50\\_1351748785.pdf](http://idl-envi.co.kr/callcenter/tips/file/50_1351748785.pdf)